// File: contracts/DynamicDescriptions.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.7;

import "@openzeppelin/contracts/utils/Strings.sol";
import "@openzeppelin/contracts/access/Ownable.sol";

/// @title BLONKS Dynamic Description Contract v3.0.0
/// @author Matto AKA MonkMatto
/// @notice This contract creates dynamic token descriptions for BLONKS.
/// @dev For flexibility, the descriptions are stored in an array of strings that are assembled with dynamic data.
/// An external contract is added for additional flexibility.
/// @custom:experimental This is an experimental contract.
/// @custom:security-contact monkmatto@protonmail.com

interface iEE {
    function assembleBasicData(
        address _account
    ) external view returns (string[] memory);

    function isComposable(address _account) external view returns (bool);
}

contract DynamicDescriptions is Ownable {
    using Strings for string;

    address public EEcontract;
    bool public EEcontractActive;

    string private EElinkBase;
    string[] private descText;

    function setDescriptionText(string[] memory _text) external onlyOwner {
        for (uint256 i = 0; i < _text.length; i++) {
            descText.push(_text[i]);
        }
    }

    function getDescriptionText() external view returns (string[] memory) {
        return descText;
    }

    function setEEcontract(address _EEcontract) external onlyOwner {
        EEcontract = _EEcontract;
    }

    function setEElinkBase(string memory _EElinkBase) external onlyOwner {
        EElinkBase = _EElinkBase;
    }

    function toggleEEcontractActive() external onlyOwner {
        EEcontractActive = !EEcontractActive;
    }

    function updateDescriptionTextIndex(
        uint256 _index,
        string memory _text
    ) external onlyOwner {
        descText[_index] = _text;
    }

    function buildDynamicDescription(
        address _ownerAddy,
        uint256 _shapeshiftCount,
        string memory _collectionDescription,
        string memory _shifterName,
        uint256 _shifterActive,
        uint256 _shifterMax
    ) external view returns (string memory) {
        string memory shapeshiftDetail = _shapeshiftCount > 0
            ? string(
                abi.encodePacked(descText[9], Strings.toString(_shapeshiftCount), descText[10])
            )
            : "";
        string memory dynamicText = EEcontractActive && iEE(EEcontract).isComposable(_ownerAddy)
            ? _incorporateEE(_ownerAddy)
            : descText[0];
        dynamicText = string(
            abi.encodePacked(
                dynamicText,
                descText[11],
                descText[12],
                _shifterName
            )
        );
        dynamicText = string(
            abi.encodePacked(
                dynamicText,
                shapeshiftDetail,
                descText[13],
                Strings.toHexString(uint160(_ownerAddy), 20),
                descText[14]
            )
        );
        dynamicText = string(
            abi.encodePacked(
                dynamicText,
                _shifterName,
                descText[15],
                Strings.toString(_shifterActive),
                descText[16],
                Strings.toString(_shifterMax),
                descText[8]
            )
        );
        dynamicText = string(
            abi.encodePacked(
                dynamicText,
                descText[17],
                _collectionDescription,
                descText[18]
            )
        );
        return dynamicText;
    }

    function _incorporateEE(
        address _ownerAddy
    ) internal view returns (string memory) {
        string[] memory EEbasicData = iEE(EEcontract).assembleBasicData(
            _ownerAddy
        );
        string memory EEdynamicText;
        if (
            bytes(EEbasicData[0]).length +
                bytes(EEbasicData[1]).length +
                bytes(EEbasicData[2]).length >
            0
        ) {
            EEdynamicText = descText[1];
        }
        if (bytes(EEbasicData[0]).length > 0) {
            EEdynamicText = string(
                abi.encodePacked(EEdynamicText, descText[2], EEbasicData[0])
            );
        }
        if (bytes(EEbasicData[1]).length > 0) {
            EEdynamicText = string(
                abi.encodePacked(EEdynamicText, descText[3], EEbasicData[1])
            );
        }
        uint256 priorityLink = uint256(uint8(bytes(EEbasicData[5])[0])) - 48;
        if (bytes(EEbasicData[priorityLink + 2]).length > 0) {
          EEdynamicText = string(abi.encodePacked(EEdynamicText, descText[priorityLink + 4], EEbasicData[priorityLink + 2]));
        }
        EEdynamicText = string(
            abi.encodePacked(
                EEdynamicText,
                descText[7],
                EElinkBase,
                Strings.toHexString(uint160(_ownerAddy), 20),
                descText[8]
            )
        );
        return EEdynamicText;
    }
}
// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}